# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=tasty-hspec
pkgname=haskell-tasty-hspec
pkgver=1.2.0.3
pkgrel=18
pkgdesc="Hspec support for the Tasty test framework"
url="https://github.com/mitchellwrosen/tasty-hspec"
license=('BSD')
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-hspec' 'haskell-hspec-core' 'haskell-quickcheck' 'haskell-tasty'
         'haskell-tasty-smallcheck' 'haskell-tasty-quickcheck')
makedepends=('ghc')
source=(https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz)
sha512sums=('7b1c23a3ad2f3392fa0200bcf193a2c6c743f1729208e308173ab4b11c4b5358aaf5146eff8a37a25934d7b2d01847a7de4e82fb0b642cc57a9fc46f373f77b3')

build() {
    cd $_hkgname-$pkgver

    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $_hkgname-$pkgver
    runhaskell Setup test --show-details=direct
}

package() {
    cd $_hkgname-$pkgver

    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
